Calcul Formel et Sciences de la Matière — Janvier 2007
Durée 3 heures — Documents et calculatrices interdits

Prenez le temps de lire ce sujet. Les questions sont indépendantes pour la plupart.

1 Moindres carrés (4 points)

Dans la salle d'attente du service après–vente d'une grande surface spécialisée en ameublement, on peut lire un écriteau du type suivant :

Estimez votre temps d'attente :
Comptez le nombre de personnes qui attendent avant vous.
Appelez n ce nombre.
Votre temps d'attente est approximativement $3\,n+1$ minutes.

Une grande surface concurrente a décidé de mettre au point une formule du même type. Les temps d'attente de plusieurs clients ont été chronométrés. Les voici.

client	tps d'attente (en minutes)	nbre de clients qui précèdent
A	5	1
B	8	2
C	9	2
D	14	3
E	17	4

Le responsable du service après–vente vous demande d'établir une formule du type

$$\textit{minutes d'attente} = a \times [\textit{nbre de clients qui précèdent}] + b$$

qui corresponde le mieux aux mesures, au sens des moindres carrés.

Question 1 [1 pt]. Quelles sont les inconnues du problème ?

SOLUTION. a et b.

Question 2 [2 pts]. Au problème énoncé ci–dessus correspond un système d'équations linéaires surdéterminé. Écrire ce système sous forme matricielle (ne pas le résoudre).

SOLUTION.

$$\begin{pmatrix} 1 & 1 \\ 2 & 1 \\ 2 & 1 \\ 3 & 1 \\ 4 & 1 \end{pmatrix} \begin{pmatrix} a \\ b \end{pmatrix} = \begin{pmatrix} 5 \\ 8 \\ 9 \\ 14 \\ 17 \end{pmatrix}.$$

Question 3 [1 pt]. Les inconnues du problème sont les solutions d'un système exactement déterminé (un système carré) qui se déduit du système précédent. Comment s'obtient ce système (donner seulement la formule, ne pas calculer ce système, ne pas le résoudre) ?

SOLUTION. En notant $X\,a = y$ le système ci–dessus, il s'agit du système ${}^tX\,X\,a = {}^tX\,y$.

2 Équations différentielles (9 points)

2.1 Une équation en une variable

On considère l'équation différentielle ordinaire suivante :

$$\dot{x} = (x+1)(x-3).$$

Question 4 [1 pt]. Donner les solutions constantes de l'équation ainsi que leur nature (dessiner la ligne de phase).

SOLUTION. Les solutions constantes sont $x(t) = -1$ (attractive) et $x(t) = 3$ (répulsive).

Question 5 [1 pt]. Appliquer un pas de la méthode d'Euler à partir d'une condition initiale $x(0) = 2$ et un pas de temps $h = 1/10$.

SOLUTION. On trouve $17/10$.

Question 6 [2 pts]. Donner une commande MAPLE qui affecte cette équation à une variable *edo* et qui l'intègre numériquement.

```
edo := diff (x(t),t) = (x(t)+1)*(x(t)-3);
dsolve ({edo, x(0)=2}, x(t), numeric};
```

2.2 Deux équations en deux variables

On considère la matrice

$$A = \begin{pmatrix} 1 & 4 \\ 2 & 3 \end{pmatrix}$$

et le système différentiel linéaire

$$\dot{x} = A\,x.$$

Question 7 [2 pts]. Vérifier que 5 et -1 sont les deux valeurs propres de A.

SOLUTION. On peut vérifier par exemple que le polynôme caractéristique de A s'écrit $(\lambda - 5)\,(\lambda + 1)$.

Question 8 [1 pt]. Quelle est la nature de la solution constante du système différentiel ?

SOLUTION. C'est un col.

Question 9 [2 pts]. Donner une commande MAPLE qui affecte la matrice ci–dessus à une variable A puis qui calcule les valeurs propres de A (on suppose le paquetage *LinearAlgebra* chargé en mémoire).

```
A := <<1 | 4>, <2 | 3>>;
Eigenvalues (A);
```

3 Méthode de Newton (12 points)

La méthode de Newton, étudiée en cours, permet de déterminer une racine d'une équation numérique

$$F(x) = 0.$$

Question 10 [1 pt]. Écrire la relation de récurrence de la suite de Newton pour $F(x) = x^3 - 2$. Simplifier un peu la formule.

SOLUTION. On trouve $x_{n+1} = \dfrac{2\,(x_n^3 + 1)}{3\,x_n^2}$.

Question 11 [1 pt]. Calculer x_1 à partir de $x_0 = 2$.

SOLUTION. On trouve $x_1 = 3/2$.

Le schéma de Newton étudié en cours se généralise aux systèmes de deux équations en deux variables. Le principe mathématique de cette généralisation est expliqué en section 3.1. En section 3.2, on donne une suite d'instructions MAPLE qui applique la généralisation sur un exemple. Les questions sont données en section 3.3.

Remarque : il n'est pas nécessaire de comprendre en détail le contenu de la section 3.1 pour répondre aux questions de la section 3.3.

3.1 Principe mathématique

On considère un système de deux équations de deux variables réelles :

$$F(x,\,y) = 0, \quad G(x,\,y) = 0.$$

La suite de Newton consiste à calculer une suite de points $(x_n,\,y_n)$ à partir d'un point donné $(x_0,\,y_0)$. La relation de récurrence pour le schéma de Newton en deux variables s'écrit :

$$x_{n+1} = x_n + h, \quad y_{n+1} = y_n + \ell$$

où h et ℓ sont deux réels qui sont recalculés à chaque itération. Ces réels sont en fait la solution du système d'équations suivant, qui dépend de x_n et de y_n et qu'il suffit donc de résoudre à chaque itération. La matrice des coefficients est souvent appelée la matrice *jacobienne* du système à résoudre

$F(x, y) = 0,\ G(x, y) = 0$:

$$\begin{pmatrix} \dfrac{\partial F}{\partial x}(x_n,\, y_n) & \dfrac{\partial F}{\partial y}(x_n,\, y_n) \\ \dfrac{\partial G}{\partial x}(x_n,\, y_n) & \dfrac{\partial G}{\partial y}(x_n,\, y_n) \end{pmatrix} \cdot \begin{pmatrix} h \\ \ell \end{pmatrix} = - \begin{pmatrix} F(x_n,\, y_n) \\ G(x_n,\, y_n) \end{pmatrix}.$$

Il n'est pas nécessaire de lire les explications complémentaires qui suivent pour répondre aux questions données en section 3.3. Elles sont seulement destinées à expliquer la provenance du système d'équations.

3.1.1 Explications complémentaires

Soient $(\bar{x},\, \bar{y})$ une solution de ce système et $(x_n,\, y_n)$ un couple de réels proche de $(\bar{x},\, \bar{y})$. Notons h et ℓ les différences $\bar{x} - x_n$ et $\bar{y} - y_n$ de telle sorte que $\bar{x} = x_n + h$ et $\bar{y} = y_n + \ell$. Alors, en appliquant la formule du développement limité de $F(x,\, y)$ et de $G(x,\, y)$ en $(x_n,\, y_n)$ on obtient une formule :

$$0 = F(\bar{x},\, \bar{y}) = F(x_n + h,\, y_n + \ell) = F(x_n,\, y_n) + h\, \frac{\partial F}{\partial x}(x_n,\, y_n) + \ell\, \frac{\partial F}{\partial y}(x_n,\, y_n) + \cdots$$

$$0 = G(\bar{x},\, \bar{y}) = G(x_n + h,\, y_n + \ell) = G(x_n,\, y_n) + h\, \frac{\partial G}{\partial x}(x_n,\, y_n) + \ell\, \frac{\partial G}{\partial y}(x_n,\, y_n) + \cdots$$

Comme h et ℓ sont petits, on se permet de négliger les termes cachés dans les points de suspension et on a :

$$\begin{aligned} 0 &\simeq F(x_n,\, y_n) + h\, \frac{\partial F}{\partial x}(x_n,\, y_n) + \ell\, \frac{\partial F}{\partial y}(x_n,\, y_n), \\ 0 &\simeq G(x_n,\, y_n) + h\, \frac{\partial G}{\partial x}(x_n,\, y_n) + \ell\, \frac{\partial G}{\partial y}(x_n,\, y_n)\cdot \end{aligned}$$

Transformons les $\simeq$ en $=$ et écrivons ce système matriciellement. On obtient un système de deux équations linéaires dont les inconnues sont les réels h et ℓ. Il s'agit du système donné en début de section.

3.2 Calcul interactif en MAPLE

Les commandes MAPLE suivantes montrent comment on peut appliquer la méthode de Newton décrite en section 3.1 pour déterminer une solution du système :

$$x^2 + y^2 - 10 = 0, \quad x - 3\,y = 0.$$

On peut montrer que ce système a deux solutions réelles, qui sont $(x, y) = (3, 1)$ et $(-3, -1)$.

```
> with (LinearAlgebra):
> with (VectorCalculus,Jacobian):
> F := x^2 + y^2 - 10:
```

```
> G := x - 3*y:

> M := < Jacobian ([F, G], [x, y]) | - <F, G> >;

                           [                 2    2     ]
                      M := [2 x    2 y    -x   - y  + 10]
                           [                            ]
                           [ 1     -3       -x + 3 y    ]

############## Point de départ
> x0 := 2.:
> y0 := 2.:
> M0 := subs (x=x0, y=y0, M);
                                  [4.    4.    2.]
                            M0 := [              ]
                                  [1     -3    4.]

> hl := LinearSolve (M0);
                               [ 1.37500000000000000 ]
                         hl := [                     ]
                               [-0.875000000000000000]

############## Point suivant
> x1 := x0 + hl[1];
                               x1 := 3.375000000

> y1 := y0 + hl[2];
                               y1 := 1.125000000

> M1 := subs (x=x1, y=y1, M);
                    [6.750000000    2.250000000    -2.656250000]
             M1 := [                                           ]
                    [     1             -3              0.     ]

> hl := LinearSolve (M1);
                               [-0.354166666666666685]
                         hl := [                     ]
                               [-0.118055555555555538]

############## Point suivant
> x2 := x1 + hl[1];
                               x2 := 3.020833333

> y2 := y1 + hl[2];
                               y2 := 1.006944444
```

3.3 Questions

Question 12 [1 pt]. Quels sont les trois points $(x_n,\, y_n)$ calculés par l'itération de Newton ci–dessus ?

SOLUTION. Ce sont $(x_0, y_0) = (2, 2)$, $(x_1, y_1) = (3.375, 1.125)$, $(x_2, y_2) = (3.021, 1.007)$.

Question 13 [1 pt]. Vers quelle solution du système l'itération de Newton semble–t–elle converger ?

SOLUTION. Apparemment, $(3, 1)$.

3.3.1 Généralisation

On souhaite améliorer un peu les calculs effectués ci–dessus en utilisant une boucle et en affectant les coordonnées des points à des variables indicées x_n et y_n.

Question 14 [2 pts]. On suppose que la variable N contient un entier strictement positif. Écrire une suite d'instructions MAPLE, comportant une boucle *for*, qui calcule les N premiers points de l'itération de Newton décrite ci–dessus. Vos instructions doivent remplacer toutes les commandes données en exemple qui suivent le commentaire :

```
############## Point de départ

x[0] := 2.;
y[0] := 2.;
for i from 1 to N-1 do
    Mi := subs (x=x[i-1], y=y[i-1], M);
    hl := LinearSolve (Mi);
    x[i] := x[i-1] + hl[1];
    y[i] := y[i-1] + hl[2];
od;
```

Question 15 [2 pts]. Transformer la boucle *for* donnée à la question précédente en une boucle *while* équivalente (si vous n'avez pas répondu à la question précédente, vous pouvez expliquer le principe de la modification).

```
x[0] := 2.;
y[0] := 2.;
i := 1;
while i <= N-1 do
    Mi := subs (x=x[i-1], y=y[i-1], M);
    hl := LinearSolve (Mi);
    x[i] := x[i-1] + hl[1];
    y[i] := y[i-1] + hl[2];
    i := i+1
od;
```

3.3.2 Convergence

On souhaite améliorer encore un peu les calculs effectués en arrêtant les calculs dès que le vecteur $(h\ \ell)$ est de norme euclidienne inférieure à un $\varepsilon > 0$ donné, c'est–à–dire dès que

$$\sqrt{h^2 + \ell^2} < \varepsilon.$$

Question 16 [2 pts]. Écrire en MAPLE une fonction *norme*, paramétrée par un vecteur et qui retourne sa norme euclidienne, c'est–à–dire la racine carrée de la somme des carrés de ses coordonnées.

Votre fonction doit pouvoir s'appliquer à un vecteur de dimension quelconque.

Il est interdit d'utiliser la fonction *Norm* du paquetage *LinearAlgebra*.

```
norme := proc (V)
    local S, i;
    S := 0;
    for i from 1 to Dimension (V) do
        S := S + V[i]^2
    od;
    sqrt (S)
end:
```

Question 17 [2 pts]. Transformer la boucle *while* écrite deux questions plus haut de telle sorte que les calculs s'arrêtent dès que l'indice n atteint N ou que la norme du vecteur *hl* est inférieure à ε. On suppose la variable *epsilon* définie. Utiliser la fonction définie précédemment.

```
x[0] := 2.;
y[0] := 2.;
i := 1;

Mi := subs (x=x[i-1], y=y[i-1], M);
hl := LinearSolve (Mi);
x[i] := x[i-1] + hl[1];
y[i] := y[i-1] + hl[2];
i := i+1;

while i <= N-1 and norme (hl) > epsilon do
    Mi := subs (x=x[i-1], y=y[i-1], M);
    hl := LinearSolve (Mi);
    x[i] := x[i-1] + hl[1];
    y[i] := y[i-1] + hl[2];
    i := i+1
od;
```